

Pensieve Header: Verifying the Conway and w-Conway relations using pA.

Program copied from <http://www.math.toronto.edu/~drorbn/Talks/Sandbjerg-0810/index.php>.

```
(*WP:Wedge Product*)
WSort[expr_] := Expand[expr /. w_W => Signature[w] * Sort[w]];
WP[0, _] = WP[_ , 0] = 0;
WP[a_, b_] :=
  WSort[Distribute[a ** b] /. (c1_. * w1_W) ** (c2_. * w2_W) => c1 c2 Join[w1, w2]];

(*IM:Interior Multiplication*)
IM[{}, expr_] := expr;
IM[i_, w_W] := If[FreeQ[w, i], 0, -(-1)^Position[w, i][[1, 1]] * DeleteCases[w, i]];
IM[{is___, i_}, w_W] := IM[{is}, IM[i, w]];
IM[is_List, expr_] := expr /. w_W => IM[is, w]


(*pA on Crossings*)
pA[Xp[i_, j_, k_, L_]] := AHD[(t[i] == t[k]) (t[j] == t[L]), {i, L},
  W[j, k], W[L, i] + (t[i] - 1) W[L, j] - t[L] W[L, k] + W[i, j] + t[L] W[j, k]];
pA[Xm[i_, j_, k_, L_]] := AHD[(t[i] == t[k]) (t[j] == t[L]), {i, j}, W[k, L],
  t[j] W[i, j] - t[j] W[i, L] + W[j, k] + (t[i] - 1) W[j, L] + W[k, L]]

(*Variable Equivalences*)
ReductionRules[Times[]] = {};
ReductionRules[Equal[a_, b_]] := (# -> a) & /@ {b};
ReductionRules[eqs_Times] := Join@@(ReductionRules /@ List@@ eqs)

(*AHD:Alexander Half Densities*)
AHD[eqs_, is_, -os_, p_] := AHD[eqs, is, os, Expand[-p]];
AHD /: Reduce[AHD[eqs_, is_, os_, p_]] :=
  AHD[eqs, Sort[is], WSort[os], WSort[p /. ReductionRules[eqs]]];
AHD /: AHD[eqs1_, is1_, os1_, p1_] AHD[eqs2_, is2_, os2_, p2_] :=
  Module[{glued = Intersection[Union[is1, is2], List@@Union[os1, os2]]},
    Reduce[AHD[eqs1 * eqs2 /.
      eq1_Equal * eq2_Equal /: Intersection[List@@eq1, List@@eq2] != {} => Union[eq1, eq2],
      Complement[Union[is1, is2], glued], IM[glued, WP[os1, os2]], IM[glued, WP[p1, p2]]]]]]

(*pA on Circuit Diagrams*)
pA[cd_CircuitDiagram, eqs___] := pA[cd, {}, AHD[Times[eqs], {}, W[], W[]]];
pA[cd_CircuitDiagram, done_, ahd_AHD] :=
  Module[{pos = First[Ordering[Length[Complement[List@@#, done]] & /@ cd]],
    pA[Delete[cd, pos], Union[done, List@@cd[[pos]], ahd * pA[cd[[pos]]]]];
pA[CircuitDiagram[], _, ahd_AHD] := ahd

pA[CircuitDiagram[Xp[2, 4, 3, 1], Xm[3, 4, 6, 5]]]
AHD[(t[1] == t[4] == t[5]) (t[2] == t[3] == t[6]), {1, 2},
  W[5, 6], t[1] W[1, 2] - t[1] W[1, 6] + t[1] W[2, 5] + t[1] W[5, 6]]
```



```

T0 = WSort [
  AHD [ (t[1] == t[2] == t[3] == t[4]), {1, 2}, W[4, 3], W[1, 2] - W[1, 3] + W[2, 4] + W[4, 3] ] ]
AHD [t[1] == t[2] == t[3] == t[4], {1, 2}, W[3, 4], -W[1, 2] + W[1, 3] - W[2, 4] + W[3, 4] ]

T1 = AHD [ (t[1] == t[2] == t[3] == t[4]), {1, 2}, W[3, 4], W[1, 2] - W[1, 4] + W[2, 3] + W[3, 4] ];

T2 = pA[CircuitDiagram[Xp[2, 4, 3, 1]], t[1] == t[2]]
AHD [t[1] == t[2] == t[3] == t[4], {1, 2}, W[3, 4],
  -W[1, 2] + t[1] W[1, 3] + W[1, 4] - t[1] W[1, 4] - W[2, 4] + t[1] W[3, 4] ]

T3 = pA[CircuitDiagram[Xm[1, 2, 4, 3]], t[1] == t[2]]
AHD [t[1] == t[2] == t[3] == t[4], {1, 2}, W[3, 4],
  -t[1] W[1, 2] + t[1] W[1, 3] + W[2, 3] - t[1] W[2, 3] - W[2, 4] + W[3, 4] ]

(Last /@ {T0, T1, T2, T3}) // MatrixForm

$$\begin{pmatrix} -W[1, 2] + W[1, 3] - W[2, 4] + W[3, 4] \\ W[1, 2] - W[1, 4] + W[2, 3] + W[3, 4] \\ -W[1, 2] + t[1] W[1, 3] + W[1, 4] - t[1] W[1, 4] - W[2, 4] + t[1] W[3, 4] \\ -t[1] W[1, 2] + t[1] W[1, 3] + W[2, 3] - t[1] W[2, 3] - W[2, 4] + W[3, 4] \end{pmatrix}$$


Factor[Last[T2] - Last[T3]]
(-1 + t[1]) (W[1, 2] - W[1, 4] + W[2, 3] + W[3, 4])

Factor[Last[T2] - Last[T0]]
(-1 + t[1]) (W[1, 3] - W[1, 4] + W[3, 4])

```

Moral: There doesn't seem to be a w-Conway relation.